



Take special care when using Justified Right with an input field. As your data is keyed into the input field, it will be left-justified, and will not be positioned in the field as you might expect when using Justified Right. Justified Right is only used to position the field in the display. When the screen description is accepted, the field, exactly as the user keyed it, will reside in the target field. I suggest that Justified Right clause be used sparingly, if at all.

To and From can be used together for the same screen description elementary item. Doing so causes one item from the Data Division to be displayed while accepting data into a different data item. This approach can be used to preserve the original display field.

```
000032 03 Line 6 Column 22 Pic x(5) From Item-Description
000033 To New-Item-Description.
```

### Using Using

When Using is specified, an update item is created. An update item displays and accepts input into the same data area. Any changes are shown when the screen description is next displayed. An update item may have the same special clauses as input and output items. Using allows you to use a single data item for display and update by the user.

## Special-Names Paragraph

When using the Screen Section, you sometimes have to capture and set the cursor location. In addition, specific function keys can be activated that may be detected by your program. If no function keys are activated, the only key that can terminate the Accept of a screen description is the Enter key.

To capture the cursor and the function keys, you need to make entries in the Special-Names paragraph of the Configuration Section. These entries relate the actual cursor position and function key status to Working-Storage data items that you can reference in the program. The two special names you will assign are Cursor and Crt Status. Cursor is the position of the cursor. When you make the Special-Names entry, you are specifying a data item that contains the row and column of the position of the cursor. When a screen definition is displayed, the cursor appears at the field, with the starting position closest to, but not less than, the row and column specified in the cursor field. The field referenced by the Cursor special name must be either four or six characters long. If four characters, the first two are the row and the last two are the column of the cursor position. If the field is six characters long, then the first three correspond to the row and the last three to the column.

```
000005 Configuration Section.
000006 Special-Names.
000007 Cursor Is Cursor-Position.
000008 Source-Computer. IBM-PC.
000009 Object-Computer. IBM-PC.
000010 Data Division.
000011 Working-Storage Section.
000012 01 Cursor-Position.
000013 03 Cursor-Row Pic 9(2) value 1.
000014 03 Cursor-Column Pic 9(2) value 1.
```

Line 6 starts the Special-Names paragraph. Only the last item in the paragraph should be followed by a period. The Special-Names entry, Cursor, for example, starts in Area B (column 12).



Some compilers differ in the area of the specifications for the Cursor Special-Names entry. The cursor position field for the VMS COBOL compiler DECAlpha systems may be either four or five positions long. If four positions, the first two positions are the row and the last two the column. If five positions long, the last three are the column number. If you are not using the Fujitsu compiler, see the language reference provided with your compiler to determine the proper values for the Cursor Special-Names entry.

The other Special-Names entry associated with the Screen Section is Crt Status.

The field assigned to the Crt Status special name is three characters long. The first two positions provide codes that correspond to the reason for the termination of the Accept. The system uses the third position for internal housekeeping and should not be referenced.

It is useful to define this status value as a Group Level item with three subordinate elementary items corresponding to the three individual return characters.

```
000013 01 Keyboard-Status.
000014 03 Accept-Status Pic 9.
000015 03 Function-Key Pic X.
000016 03 System-Use Pic X.
```

The first character, Accept-Status, contains a 0 if the Accept is terminated normally, either by the Enter key being pressed or by the last field in the screen definition being filled when the Auto clause is specified. In this case, the second character, Accept-Status, contains either a 0 or a 1. A value of 0 means that the user terminated the Accept by pressing Enter. A value of 1 means that the user filled the last field of the screen and the Accept was terminated because the Auto clause was specified.

Accept-Status has a value of 1 or 2 if the accept statement is terminated by the press of a function key. In this case, the second field contains a coded value corresponding to the function key that is pressed. A value of 1 in the Accept-Status field indicates that a default function key terminated the Accept, whereas a value of 2 indicates a user-defined function key.



The size and meaning of the Crt-Status data item depend on the COBOL compiler. If you are not using the Fujitsu compiler, check your documentation for the appropriate size and meanings for the different values.

```
000005 Configuration Section.
000006 Special-Names.
000007   Crt-Status is Keyboard-Status
000008   Cursor is Cursor-Position.
000009 Source-Computer. IBM-PC.
000010 Object-Computer. IBM-PC.
000011 Data Division.
000012 Working-Storage Section.
000013 01 Keyboard-Status.
000014   03 Accept-Status Pic 9.
000015   03 Function-Key Pic X.
000016   03 System-Use Pic X.
000017 01 Cursor-Position.
000018   03 Cursor-Row Pic 9(2) Value 1.
000019   03 Cursor-Column Pic 9(2) Value 1.
```



Notice the single period after the statements in the Special-Names paragraph. If you need to specify any other items in the Special-Names paragraph, remember to use a single period after the last item only. Additionally, some compilers are sensitive to the order of items listed in the Configuration Section. If you have trouble compiling the program under a different compiler, try placing the Special-Names paragraph after the Source-Computer and Object-Computer paragraphs.

## Using the Screen Section in a Program

The simple data entry screen in the following example gathers information about the different sellers in a consignment store. Before designing any screens, you need to consider the requirements. What kind of data is to be collected? How is it to be displayed? What do you want the screen to look like? The program displays a screen and then waits for

the user to enter the data. Some default values are provided so that the user does not need to key everything.

First, decide which items you need to track and the size you want to assign to them. Be very careful to make the fields large enough without being wasteful. COBOL programmers are always mindful of future maintenance needs in their programs. Create yours with that in mind, and if you or other programmers have to modify the program, the task will be easy.

The program tracks the following items for each tenant, using the specified field types and lengths. The default value to assign is also listed.

- Last Name—Alphanumeric 25 characters
- First Name—Alphanumeric 15 characters
- Middle Name—Alphanumeric 10 characters
- Address Line 1—Alphanumeric 50 characters
- Address Line 2—Alphanumeric 50 characters
- City—Alphanumeric 40 characters
- State or Country—Alphanumeric 20 characters
- Postal Code—Alphanumeric 15 characters
- Home Telephone—Alphanumeric 20 characters
- Work Telephone—Alphanumeric 20 characters
- Other Telephone—Alphanumeric 20 characters
- Start Date—Numeric eight digits, formatted MM/DD/YYYY
- Last Rent Paid Date—Numeric eight digits, formatted MM/DD/YYYY
- Next Rent Due Date—Numeric eight digits, formatted MM/DD/YYYY
- Rent Amount—Numeric six digits, two decimal positions, default \$50.00
- Consignment Percentage—Numeric three digits, default 40

Take special notice of the extra space in the Postal Code and Telephone Number fields. Also, notice that the dates are eight digits long, even though with the slashes they fill 10 display positions.

The required fields are First Name, Last Name, Home Telephone, Start Date, Rent Amount, and Consignment Percentage. Try to format the screen clearly and neatly, using literals to title the various fields. Make the entry fields reverse video to differentiate them from the screen literals.

The screen requires a title describing its purpose and a fancy store name. Use Darlene's Treasures. Listing 4.2 is one way to code the Screen Section.

Key the following program into the editor and name it **Chap04C.Cob**.

#### Listing 4.2 SCREEN SECTION DEMONSTRATION

```
000001 @OPTIONS MAIN
000002 Identification Division.
000003 Program-Id. Chap04C.
000004* Data entry Screen
000005 Environment Division.
000006 Configuration Section.
000007 Special-Names.
000008 Crt Status is Keyboard-Status
000009 Cursor is Cursor-Position.
000010 Source-Computer. IBM-PC.
000011 Object-Computer. IBM-PC.
000012 Data Division.
000013 Working-Storage Section.
000014 01 Keyboard-Status.
000015 03 Accept-Status Pic 9.
000016 03 Function-key Pic X.
000017 03 System-Use Pic X.
000018 01 Cursor-Position.
000019 03 Cursor-Row Pic 9(2) Value 1.
000020 03 Cursor-Column Pic 9(2) Value 1.
000021 01 Screen-Items.
000022 03 Last-Name Pic X(25) Value Spaces.
000023 03 First-Name Pic X(15) Value Spaces.
000024 03 Middle-Name Pic X(10) Value Spaces.
000025 03 Address-Line-1 Pic X(50) Value Spaces.
000026 03 Address-Line-2 Pic X(50) Value Spaces.
000027 03 City Pic X(40) Value Spaces.
000028 03 State-or-Country Pic X(20) Value Spaces.
000029 03 Postal-Code Pic X(15) Value Spaces.
000030 03 Home-Phone Pic X(20) Value Spaces.
000031 03 Work-Phone Pic X(20) Value Spaces.
000032 03 Other-Phone Pic X(20) Value Spaces.
000033 03 Start-Date Pic 9(8) Value Zeros.
000034 03 Last-Rent-Paid-Date Pic 9(8) Value Zeros.
000035 03 Next-Rent-Due-Date Pic 9(8) Value Zeros.
000036 03 Rent-Amount Pic 9(4) Value 50.00.
000037 03 Consignment-Percent Pic 9(3) Value 40.
000038 Screen Section.
000039 01 Data-Entry-Screen
000040 Blank Screen, Auto
000041 Foreground-Color is 7,
000042 Background-Color is 1.
```

```
000043 03 Line 01 Column 30 Value "Darlene's Treasures"
000044 Highlight Foreground-Color 4 Background-Color 1.
000045 03 Line 03 Column 30 Value "Tenant Entry Program"
000046 Highlight.
000047*
000048 03 Line 5 Column 01 Value "Name, Last: ".
000049 03 Line 5 Column 13 Pic X(25) Using Last-Name
000050 Reverse-Video Required.
000051 03 Line 5 Column 39 Value "First: ".
000052 03 Line 5 Column 46 Pic X(15) Using First-Name
000053 Reverse-Video Required.
000054 03 Line 5 Column 62 Value "Middle: ".
000055 03 Line 5 Column 70 Pic X(10) Using Middle-Name
000056 Reverse-Video.
000057*
000058 03 Line 6 Column 01 Value "Address 1: ".
000059 03 Line 6 Column 15 Pic X(50) Using Address-Line-1
000060 Reverse-Video.
000061*
000062 03 Line 7 Column 01 Value "Address 2: ".
000063 03 Line 7 Column 15 Pic X(50) Using Address-Line-2
000064 Reverse-Video.
000065*
000066 03 Line 8 Column 01 Value "City: ".
000067 03 Line 8 Column 15 Pic X(40) Using City
000068 Reverse-Video.
000069*
000070 03 Line 9 Column 01 Value "Country/State: ".
000071 03 Line 9 Column 15 Pic X(20) Using State-Or-Country
000072 Reverse-Video.
000073 03 Line 9 Column 36 Value "Postal Code: ".
000074 03 Line 9 Column 50 Pic X(15) Using Postal-Code
000075 Reverse-Video.
000076*
000077 03 Line 11 Column 01 Value "Phone/Home: ".
000078 03 Line 11 Column 13 Pic X(20) Using Home-Phone
000079 Reverse-Video.
000080 03 Line 11 Column 34 Value "Work: ".
000081 03 Line 11 Column 41 Pic X(20) Using Work-Phone
000082 Reverse-Video.
000083*
000084 03 Line 12 Column 06 Value "Other: ".
000085 03 Line 12 Column 13 Pic X(20) Using Other-Phone
000086 Reverse-Video.
000087*
000088 03 Line 14 Column 01 Value "Start Date: ".
000089 03 Line 14 Column 13 Pic 99/99/9999 Using Start-Date
000090 Reverse-Video.
000091 03 Line 14 Column 24 Value "Last Paid Date: ".
```

continues

## LISTING 4.2 CONTINUED

```

000092      03 Line 14 Column 40 Pic 99/99/9999 Using Last-Rent-Paid-Date
000093      Reverse-Video.
000094      03 Line 14 Column 50 Value "Next Rent Due on: "
000095      03 Line 14 Column 68 Pic 99/99/9999 Using Next-Rent-Due-Date
000096      Reverse-Video.
000097      03 Line 15 Column 01 Value "Rent Amount: "
000098      03 Line 15 Column 14 Pic Z,ZZ,99 Using Rent-Amount.
000099      03 Line 16 Column 01 Value "Consignment Percent: "
000100      Reverse-Video.
000101      03 Line 16 Column 22 Pic ZZ9 Using Consignment-Percent
000102      Reverse-Video.
000103      Procedure Division.
000104      Chapter04c-Start.
000105      display Data-Entry-Screen.
000106      accept Data-Entry-Screen.
000107      Stop Run.

```

Notice the comment lines (indicated by an \* in column 7) that separate the code and make the program more readable. COBOL also tolerates plain blank lines. The fields are grouped so that the text literal appears in the screen definition before its associated field. Look at the Value clauses in use and the special screen colors. Pay special attention to the way that the subordinate data items override the attributes of the higher levels. If most of a screen is to be one color, you can code that color at the major Group Level and then override that color for individual fields at the subgroup level or even at the Elementary item Level.



When you key this program and compile it, you are liable to have typographical errors. Now is a good time to get used to correcting these errors from the compile listing. Remember that you can position the cursor on the first character of an error line and press F11 to jump to the editor screen, where you are automatically positioned at the source line that is in error. Although your compile listing may seem to disappear at this point, it hasn't. You can make your change, realizing that an error, such as a missing period on one line, can cause errors to be reported on other lines that, in fact, are correct. After making the change, you can save the program and then exit the editor to return to the compile listing, or you can minimize the edit window to see the compile listing. Another method is to click on the Window menu and select the program file you're working on. Positioning the cursor on the next error and pressing F11 repositions the cursor in the source edit window. Maximizing the edit window then displays the line in error. Alternatively, you can choose to tile the two windows.



When you save your program for the first time, make sure to specify the entire name of the program file, including the .COB file extension. If you fail to do so, you may not see your program when you try to reopen it. If that happens, rename the file to have the .COB file extension.

FIGURE 4.2

Chapter04c screen  
image.

A better, less verbose way to code the preceding Screen Section is to organize fields with the same display characteristics under a single group. This way, elements such as Reverse-Video don't need to be coded for each elementary item. The following example shows another way to code this Screen Section.

```

000038 Screen Section.
000039 01 Data-Entry-Screen
000040      Blank Screen, Auto
000041      Foreground-Color is 7,
000042      Background-Color is 1.
000043*
000044      03 Screen-Literal-Group.
000045          05 Line 01 Column 30 Value "Darlene's Treasures"
000046          Highlight Foreground-Color 4 Background-Color 1.
000047          05 Line 03 Column 30 Value "Tenant Entry Program"
000048          Highlight.
000049          05 Line 5 Column 01 Value "Name, Last: "
000050          05 Line 5 Column 39 Value "First: "
000051          05 Line 5 Column 62 Value "Middle: "
000052          05 Line 6 Column 01 Value "Address 1: "
000053          05 Line 7 Column 01 Value "Address 2: "
000054          05 Line 8 Column 01 Value "City: "

```

```

000055 05 Line 9 Column 01 Value "Country/State: "
000056 05 Line 9 Column 36 Value "Postal Code: "
000057 05 Line 11 Column 01 Value "Phone/Home: "
000058 05 Line 11 Column 34 Value "Work: "
000059 05 Line 12 Column 06 Value "Other: "
000060 05 Line 14 Column 01 Value "Start Date: "
000061 05 Line 14 Column 24 Value "Last Paid Date: "
000062 05 Line 14 Column 50 Value "Next Rent Due on: "
000063 05 Line 15 Column 01 Value "Rent Amount: "
000064 05 Line 16 Column 01 Value "Consignment Percent: "
000065 03 Required-Reverse-Group Reverse-Video Required.
000066 05 Line 5 Column 13 Pic X(25) Using Last-Name.
000067 05 Line 5 Column 46 Pic X(15) Using First-Name.
000068*
000069 03 Reverse-Video-Group Reverse-Video.
000070 05 Line 5 Column 70 Pic X(10) Using Middle-Name.
000071 05 Line 6 Column 15 Pic X(50) Using Address-Line-1.
000072 05 Line 7 Column 15 Pic X(50) Using Address-Line-2.
000073 05 Line 8 Column 15 Pic X(40) Using City.
000074 05 Line 9 Column 15 Pic X(20) Using State-Or-Country.
000075 05 Line 9 Column 50 Pic X(15) Using Postal-Code.
000076 05 Line 11 Column 13 Pic X(20) Using Home-Phone.
000077 05 Line 11 Column 41 Pic X(20) Using Work-Phone.
000078 05 Line 12 Column 13 Pic X(20) Using Other-Phone.
000079 05 Line 14 Column 13 Pic 99/99/9999 Using Start-Date.
000080 05 Line 14 Column 40 Pic 99/99/9999
000081 Using Last-Rent-Paid-Date.
000082 05 Line 14 Column 68 Pic 99/99/9999
000083 Using Next-Rent-Due-Date.
000084 05 Line 15 Column 14 Pic Z,ZZZ.99 Using Rent-Amount.
000085 05 Line 16 Column 22 Pic ZZ9 Using Consignment-Percent.

```

Run the program and experiment with it. Notice that you cannot leave the Last Name field by tabbing or by pressing the Enter key until you key some data. You must enter some data because Last Name is a required field. However, the Enter key does work after some data is keyed into the Last Name field, although First Name is also a required field. The Required attribute is in effect only while the cursor is on a field that is required. Once the requirement is satisfied for that field, Enter or Tab will work.

Also, note that only numbers may be entered in the date fields. Try to key in some letters and see what happens. You may key the slashes or omit them as you desire; the fields are always formatted with the slashes in the proper positions.

## Summary

In this hour, you learned

- About the Screen Section and its development
- How to create a screen definition
- How to apply an edit pattern to a field used in a screen definition
- How to use input, output, and update fields
- How to use Special-Names entries to get and control the cursor position and detect function keys
- Efficient ways of coding a screen definition

## Q&A

- Q What are the different elements of the Screen Section?**
- A A Screen Section is made up of one or more screen definitions. A screen definition consists of screen literals and input, output, and update fields.
- Q What is the difference between using Using and defining a screen element with both From and To fields?**
- A Specifying Using causes a field to be displayed and then updated when the user keys data into the field. Specifying From and To causes data to be displayed from one field but accepted into another.
- Q How can the position of the cursor be determined?**
- A As you write more complex programs, you will need to know which field the cursor was on last and how to position it there. When using a Screen Section, you use the Special-Names paragraph of the Configuration Section to specify a field in Working-Storage for tracking the cursor position.
- Q Can more than one screen definition be present in a program?**
- A Yes. You can specify multiple screen definitions in a Screen Section. Start each new definition with a new level 01 group.

## Workshop

To help reinforce your understanding of the material presented in this hour, refer to the section "Quiz and Exercise Questions and Answers" that can be found on the CD. This section contains quiz questions and exercises for you to complete, as well as the corresponding answers.